1 up

X-552-72-9

PREPRINT

NASA TM X- 65885

GENERATION OF THE INVARIANT COEFFICIENTS OF THE CHARACTERISTIC POLYNOMIAL FOR AN n × n MATRIX

P. R. BEAUDET
J. L. MAURY

JANUARY 1972

GSFC — GODDARD SPACE FLIGHT CENTER —
GREENBELT, MARYLAND

17p

X-552-72-9

GENERATION OF THE INVARIANT COEFFICIENTS OF THE CHARACTERISTIC POLYNOMIAL FOR AN $n \times n$ MATRIX

P. R. Beaudet
Computer Science Corporation

J. L. Maury
Goddard Space Flight Center

January 1972

Goddard Space Flight Center
Greenbelt, Maryland

PRECEDING PAGE BLANK NOT FILMED

CONTENTS

GENERATION OF THE INVARIANT COEFFICIENTS

OF THE CHARACTERISTIC POLYNOMIAL

FOR AN $n \times n$ MATRIX

1. Introduction

Often in the theories of numerical stability, one is seeking roots to a characteristic polynomial, which in the case of the predictor with iterative correction method of numerical integration, are eigenvalues of a matrix whose elements depend on the coefficients used in the integration process.

It is the purpose of this document to explicitly display the characteristic polynomial in terms of the elements of the characteristic matrix and code the same for future numerical analysis.

2. Analysis

Let a_{ij} be the elements of an $n \times n$ square matrix; and let the characteristic polynomial in λ be defined as the eigenvalue determinant

$$D(\lambda) = |a_{ij} - \lambda \delta_{ij}|$$

The explicit representation of this polynomial is

$$D(\lambda) = \sum_{\ell=0}^{n} b_\ell \lambda^\ell$$

where the coefficients, b_ℓ, can be obtained via Taylor's expression as

$$b_\ell = \frac{1}{\ell!} \left. \frac{d^\ell D}{d\lambda^\ell} \right|_{\lambda=0}$$

The plan for obtaining an explicit representation of b_ℓ will be to differentiate $D(\lambda)$ from its explicit representation as a sum of products of the elements $[a_{k,j} - \lambda \delta_{k,j}]$.

Consider the representation

$$D = \sum_{P_n\{j_i \mid i=1,2\cdots n\}} \epsilon_j \prod_{k=1}^{n} [a_{i_k, j_k} - \lambda \delta_{i_k, j_k}]$$

where the sum is over all permutations of the n elements $j_1, j_2 \ldots j_n$; ϵ_j is ± 1 depending on whether the permutation is even or odd, and i_k is the ordered sequence $\{1, \ldots n\}(i_k \equiv k)$; and the set $\{j_i \mid i = 1, 2, \ldots n\}$ are the same set of numbers, $\{1, 2 \ldots n\}$, and j_k is the k^{th} element of the permutation.

The coefficient b_0 is obtained by setting $\lambda = 0$ in $D(\lambda)$.

$$b_0 = \sum_{P_n\{j_i \mid i=1,2\cdots n\}} \epsilon_j \prod_{k=1}^{n} [a_{i_k, j_k}]$$

In accordance with the plan, we differentiate $D(\lambda)$ with respect to λ and obtain

$$\frac{dD}{d\lambda} = - \sum_{k'=1}^{n} \sum_{P_n\{j_i \mid i=1,2\cdots n\}} \epsilon_j \delta_{k', j_{k'}} \prod_{\substack{k=1 \\ \neq k'}}^{n} [a_{i_k j_k} - \lambda \delta_{i_k, j_k}].$$

Now only those permutations for which $j_{k'} = k'$ will be non-zero, and since $j_{k'}$ does not explicitly appear in the product, we may write

$$\frac{dD}{d\lambda} = - \sum_{k'=1}^{n} \sum_{P_{n-1}\{j_i \mid i=1,2\cdots n \neq k'\}} \epsilon_j \prod_{k=1}^{n-1} [a_{i_k, j_k} - \lambda \delta_{i_k, j_k}]$$

where now the set j_i excludes k' and the sequence i_k is the ordered permutation of the set of numbers $\{j_i\}$. Meaning to ϵ_j is obtained by recognizing that the ordered permutation has $\epsilon_j = +1$.

We proceed now to take other derivatives, each time we obtain a negative sum over new k's different from previous k's and the set j_i excludes the numbers taken by the set of k's in the sums. Also, the product sequence i_k becomes the ordered sequence of the j_k set. We obtain for example the second derivative as

$$\frac{d^2 D}{d\lambda^2} = (-1)^2 \sum_{k^{(1)}=1}^{n} \sum_{k^{(2)} \neq k^{(1)}} \sum_{P_{n-2}\{j_i \mid i=1\cdots n \neq k^{(1)}, k^{(2)}\}} \epsilon_j \prod_{k=1}^{n-2} [a_{i_k j_k} - \lambda \delta_{i_k, j_k}].$$

In general, the ℓ^{th} derivative becomes

$$\frac{d^\ell D}{d\lambda^\ell} = (-1)^\ell \sum_{k^{(1)}=1}^{n} \sum_{k^{(2)} \neq k^{(1)}} \cdots \sum_{k^{(\ell)} \neq k^{(\ell-1)}, \cdots k^{(1)}}$$

$$\sum_{P_{n-\ell}\{j_i \mid i=1, \cdots n \neq k^{(1)} \cdots k^{(\ell)}\}} \epsilon_j \prod_{k=1}^{n-\ell} [a_{i_k, j_k} - \lambda \delta_{i_k, j_k}].$$

Now, since only the set $\{k^{(1)} \ldots k^{(\ell)}\}$, and not its ordered sequence enters into the summand, many terms in the multiple sum on the k's will give rise to the same contribution. In particular, since $\{k^{(1)} \ldots k^{(\ell)}\}$ are distinct, there are exactly $\ell!$ ways to obtain the sequence $k^{(1)} \ldots k^{(\ell)}$ from a particular combination of ℓ numbers. As a consequence, the multiple sum on k's can be written as $\ell!$ times a sum over all combinations of ℓ numbers taken out of the n numbers, 1, 2 . . . n. Since the permutation $P_{n-\ell}$ are over the remaining elements, we could just as well choose all combinations of $n-\ell$ elements out of the n numbers and permute over this set. Explicitly, we obtain

$$\frac{d^\ell D}{d\lambda^\ell} = (-1)^\ell \ell! \sum_{C^n_{n-\ell}} \sum_{P_{n-\ell}} \epsilon_j \prod_{k=1}^{n-\ell} [a_{i_k, j_k} - \lambda \delta_{i_k, j_k}]$$

where

$\sum_{C^n_{n-\ell}}$ is the sum over all combinations of $n-\ell$ numbers out of the set of numbers 1, 2 . . . n;

$\sum_{P_{n-\ell}}$ is the sum over all permutations of the $n-\ell$ numbers obtained from each of the preceeding combinations.

ϵ_j is +1 if the sequence in the permutation is even (The ordered sequence has $\epsilon_j = 1$) and is -1 if the sequence is odd.

And the set i_k is the ordered sequence $\{j_k\}$ of the $n-\ell$ numbers from $C^n_{n-\ell}$.

The explicit representation of b_ℓ is obtained by setting $\lambda = 0$ and we obtain

$$b_\ell = (-1)^\ell \sum_{C^n_{n-\ell}} \sum_{P_{n-\ell}} \epsilon_j \prod_{k=1}^{n-\ell} [a_{i_k, j_k}]$$

3. Computer Program

The program implementing this formulation for b_ℓ consists of a MAIN routine and two subroutines, COMBIN and PERMUT. MAIN reads in a single integer N ($2 \leq N \leq 9$), the order of the matrix; then, on the basis of N creates the coefficients $b_1, b_2, \ldots, b_N$. (b_0 is not created since this is merely the determinant of the matrix $[a_{ij}]$). The coefficients are created and printed in literal, FORTRAN format. A symbolic FORTRAN program similar to the printed version can be had as a JOB CONTROL LANGUAGE option.

For an orderly generation of the coefficients, MAIN relies on the two subroutines COMBINE and PERMUT. COMBIN produces, on $\binom{n}{m}$ successive calls all combinations of the n items 1, 2, . . ., n taken m at a time. The combinations are generated in ascending order. (i.e., 1, 2, 3, < 1, 2, 4 < 1, 2, 5 < 1, 3, 4 etc.)

PERMUT produces on m! successive calls all permutations of m items and the sign of the permutation. PERMUT requires that the items to be permuted be in ascending order on the first call. With this requirement we can define each successive permutation as positive or negative obviating the time consuming chore of checking the result of a permutation.

As a matter of interest and as an indication of the practical limits on using an explicit representation of these coefficients, Table I gives the number of terms in each coefficient for various orders of n.

Table 1

NUMBER OF TERMS IN EACH COEFFICIENT OF LAMBDA FOR N X N MATRICES, N=1,10

	1 X 1	2 X 2	3 X 3	4 X 4	5 X 5	6 X 6	7 X 7	8 X 8	9 X 9	10 X 10
LAMBDA(10)										1
LAMBDA(9)									1	10
LAMBDA(8)								1	9	90
LAMBDA(7)							1	8	72	720
LAMBDA(6)						1	7	56	504	5040
LAMBDA(5)					1	6	42	336	3024	30240
LAMBDA(4)				1	5	30	210	1680	15120	151200
LAMBDA(3)			1	4	20	120	840	6720	60480	604800
LAMBDA(2)		1	3	12	60	360	2520	20160	181440	1814400
LAMBDA(1)	1	2	6	24	120	720	5040	40320	362880	3628800
TOTAL SANS LAMBDA(0)	1	3	10	41	206	1237	8660	69281	623530	6235301
TOTAL WITH LAMBDA(0)	2	5	16	65	326	1957	13700	109601	986410	9864101

Appendix A

Sample Output

LITERAL FORM OF

THE INVARIANT COEFFICIENTS OF

THE CHARACTERISTIC POLYNOMIAL FOR A

4 X 4

MATRIX

```
 F(4)=     +1
 F(3)=     +A(1,1)+A( 2,2)+A(3,3)+A(4,4)
 F(2)=     +A(3,3)*A(4,4)-A(3,4)*A(4,3)+A(2,2)*A(4,4)-A(2,4)*A(4,2)
*          +A(2,2)*A(3,3)-A(2,3)*A(3,2)+A( 1,1)*A(4,4)-A(1,4 )*A(4,1)
*          +A(1,1 )*A(3,3)-A(1,3)*A(3,1)+A(1,1)*A(2,2)-A(1,2)*A(2,1)
 F( 1)=    +A(2,2)*A(3,3)*A(4,4)-A(2,2)*A( 3,4)*A(4,3)
*          +A(2,4)*A(3,2)*A(4,3)-A(2,4)*A(3,3)*A(4,2)
*          +A(2,3)*A(3,4)*A(4,2)-A(2,3)*A(3,2)*A(4,4)
*          +A( 1,1 )*A(3,3)*A(4,4)-A(1,1)*A(3,4)*A(4,3)
*          +A(1,4)*A(3,1)*A(4,3)-A(1,4)*A(3,3)*A(4,1)
*          +A(1,3)*A(3,4)*A(4,1)-A(1,3)*A(3,1)*A(4,4)
*          +A(1,1)*A( 2,2)*A(4,4)-A(1,1)*A( 2,4)*A(4,2)
*          +A(1,4)*A(2,1)*A(4,2)-A(1,4)*A(2,2)*A(4,1)
*          +A(1,2)*A( 2,4)*A(4,1)-A(1,2)*A( 2,1)*A(4,4)
*          +A( 1,1 )*A(2,2)*A(3,3)-A(1,1)*A(2,3)*A(3,2)
*          +A(1,3)*A( 2,1)*A(3,2)-A(1,3)*A( 2,2)*A(3,1)
*          +A(1,2)*A(2,3)*A(3,1)-A(1,2)*A(2,1)*A(3,3)
```

LITERAL FORM OF

THE INVARIANT COEFFICIENTS OF

THE CHARACTERISTIC POLYNOMIAL FOR A

5 X 5

MATRIX

```
 F(5)=      +1
 F(4)=      +A(1,1)+A(2,2)+A(3,3)+A(4,4)+A(5,5)
 F(3)=      +A(4,4)*A(5,5)-A(4,5)*A(5,4)+A(3,3)*A(5,5)-A(3,5)*A(5,3)
*           +A(3,3)*A(4,4)-A(3,4)*A(4,3)+A(2,2)*A(5,5)-A(2,5)*A(5,2)
*           +A(2,2)*A(4,4)-A(2,4)*A(4,2)+A(2,2)*A(3,3)-A(2,3)*A(3,2)
*           +A(1,1)*A(5,5)-A(1,5)*A(5,1)+A(1,1)*A(4,4)-A(1,4)*A(4,1)
*           +A(1,1)*A(3,3)-A(1,3)*A(3,1)+A(1,1)*A(2,2)-A(1,2)*A(2,1)
 F(2)=      +A(3,3)*A(4,4)*A(5,5)-A(3,3)*A(4,5)*A(5,4)
*           +A(3,5)*A(4,3)*A(5,4)-A(3,5)*A(4,4)*A(5,3)
*           +A(3,4)*A(4,5)*A(5,3)-A(3,4)*A(4,3)*A(5,5)
*           +A(2,2)*A(4,4)*A(5,5)-A(2,2)*A(4,5)*A(5,4)
*           +A(2,5)*A(4,2)*A(5,4)-A(2,5)*A(4,4)*A(5,2)
*           +A(2,4)*A(4,5)*A(5,2)-A(2,4)*A(4,2)*A(5,5)
*           +A(2,2)*A(3,3)*A(5,5)-A(2,2)*A(3,5)*A(5,3)
*           +A(2,5)*A(3,2)*A(5,3)-A(2,5)*A(3,3)*A(5,2)
*           +A(2,3)*A(3,5)*A(5,2)-A(2,3)*A(3,2)*A(5,5)
*           +A(2,2)*A(3,3)*A(4,4)-A(2,2)*A(3,4)*A(4,3)
*           +A(2,4)*A(3,2)*A(4,3)-A(2,4)*A(3,3)*A(4,2)
*           +A(2,3)*A(3,4)*A(4,2)-A(2,3)*A(3,2)*A(4,4)
*           +A(1,1)*A(4,4)*A(5,5)-A(1,1)*A(4,5)*A(5,4)
*           +A(1,5)*A(4,1)*A(5,4)-A(1,5)*A(4,4)*A(5,1)
*           +A(1,4)*A(4,5)*A(5,1)-A(1,4)*A(4,1)*A(5,5)
*           +A(1,1)*A(3,3)*A(5,5)-A(1,1)*A(3,5)*A(5,3)
*           +A(1,5)*A(3,1)*A(5,3)-A(1,5)*A(3,3)*A(5,1)
*           +A(1,3)*A(3,5)*A(5,1)-A(1,3)*A(3,1)*A(5,5)
*           +A(1,1)*A(3,3)*A(4,4)-A(1,1)*A(3,4)*A(4,3)
*           +A(1,4)*A(3,1)*A(4,3)-A(1,4)*A(3,3)*A(4,1)
 F(2)=F(2)+A(1,3)*A(3,4)*A(4,1)-A(1,3)*A(3,1)*A(4,4)
*           +A(1,1)*A(2,2)*A(5,5)-A(1,1)*A(2,5)*A(5,2)
*           +A(1,5)*A(2,1)*A(5,2)-A(1,5)*A(2,2)*A(5,1)
*           +A(1,2)*A(2,5)*A(5,1)-A(1,2)*A(2,1)*A(5,5)
*           +A(1,1)*A(2,2)*A(4,4)-A(1,1)*A(2,4)*A(4,2)
*           +A(1,4)*A(2,1)*A(4,2)-A(1,4)*A(2,2)*A(4,1)
*           +A(1,2)*A(2,4)*A(4,1)-A(1,2)*A(2,1)*A(4,4)
*           +A(1,1)*A(2,2)*A(3,3)-A(1,1)*A(2,3)*A(3,2)
*           +A(1,3)*A(2,1)*A(3,2)-A(1,3)*A(2,2)*A(3,1)
*           +A(1,2)*A(2,3)*A(3,1)-A(1,2)*A(2,1)*A(3,3)
```

```
 F(1)=      +A(2,2)*A(3,3)*A(4,4)*A(5,5)-A(2,2)*A(3,3)*A(4,5)*A(5,4)
*           +A(2,2)*A(3,5)*A(4,3)*A(5,4)-A(2,2)*A(3,5)*A(4,4)*A(5,3)
*           +A(2,2)*A(3,4)*A(4,5)*A(5,3)-A(2,2)*A(3,4)*A(4,3)*A(5,5)
*           -A(2,5)*A(3,2)*A(4,3)*A(5,4)+A(2,5)*A(3,2)*A(4,4)*A(5,3)
*           -A(2,5)*A(3,4)*A(4,2)*A(5,3)+A(2,5)*A(3,4)*A(4,3)*A(5,2)
*           -A(2,5)*A(3,3)*A(4,4)*A(5,2)+A(2,5)*A(3,3)*A(4,2)*A(5,4)
*           +A(2,4)*A(3,5)*A(4,2)*A(5,3)-A(2,4)*A(3,5)*A(4,3)*A(5,2)
*           +A(2,4)*A(3,3)*A(4,5)*A(5,2)-A(2,4)*A(3,3)*A(4,2)*A(5,5)
*           +A(2,4)*A(3,2)*A(4,3)*A(5,5)-A(2,4)*A(3,2)*A(4,5)*A(5,3)
*           -A(2,3)*A(3,4)*A(4,5)*A(5,2)+A(2,3)*A(3,4)*A(4,2)*A(5,5)
*           -A(2,3)*A(3,2)*A(4,4)*A(5,5)+A(2,3)*A(3,2)*A(4,5)*A(5,4)
*           -A(2,3)*A(3,5)*A(4,2)*A(5,4)+A(2,3)*A(3,5)*A(4,4)*A(5,2)
*           +A(1,1)*A(3,3)*A(4,4)*A(5,5)-A(1,1)*A(3,3)*A(4,5)*A(5,4)
*           +A(1,1)*A(3,5)*A(4,3)*A(5,4)-A(1,1)*A(3,5)*A(4,4)*A(5,3)
*           +A(1,1)*A(3,4)*A(4,5)*A(5,3)-A(1,1)*A(3,4)*A(4,3)*A(5,5)
*           -A(1,5)*A(3,1)*A(4,3)*A(5,4)+A(1,5)*A(3,1)*A(4,4)*A(5,3)
*           -A(1,5)*A(3,4)*A(4,1)*A(5,3)+A(1,5)*A(3,4)*A(4,3)*A(5,1)
*           -A(1,5)*A(3,3)*A(4,4)*A(5,1)+A(1,5)*A(3,3)*A(4,1)*A(5,4)
*           +A(1,4)*A(3,5)*A(4,1)*A(5,3)-A(1,4)*A(3,5)*A(4,3)*A(5,1)
*           +A(1,4)*A(3,3)*A(4,5)*A(5,1)-A(1,4)*A(3,3)*A(4,1)*A(5,5)
 F(1)=F(1)  +A(1,4)*A(3,1)*A(4,3)*A(5,5)-A(1,4)*A(3,1)*A(4,5)*A(5,3)
*           -A(1,3)*A(3,4)*A(4,5)*A(5,1)+A(1,3)*A(3,4)*A(4,1)*A(5,5)
*           -A(1,3)*A(3,1)*A(4,4)*A(5,5)+A(1,3)*A(3,1)*A(4,5)*A(5,4)
*           -A(1,3)*A(3,5)*A(4,1)*A(5,4)+A(1,3)*A(3,5)*A(4,4)*A(5,1)
*           +A(1,1)*A(2,2)*A(4,4)*A(5,5)-A(1,1)*A(2,2)*A(4,5)*A(5,4)
*           +A(1,1)*A(2,5)*A(4,2)*A(5,4)-A(1,1)*A(2,5)*A(4,4)*A(5,2)
*           +A(1,1)*A(2,4)*A(4,5)*A(5,2)-A(1,1)*A(2,4)*A(4,2)*A(5,5)
*           -A(1,5)*A(2,1)*A(4,2)*A(5,4)+A(1,5)*A(2,1)*A(4,4)*A(5,2)
*           -A(1,5)*A(2,4)*A(4,1)*A(5,2)+A(1,5)*A(2,4)*A(4,2)*A(5,1)
*           -A(1,5)*A(2,2)*A(4,4)*A(5,1)+A(1,5)*A(2,2)*A(4,1)*A(5,4)
*           +A(1,4)*A(2,5)*A(4,1)*A(5,2)-A(1,4)*A(2,5)*A(4,2)*A(5,1)
*           +A(1,4)*A(2,2)*A(4,5)*A(5,1)-A(1,4)*A(2,2)*A(4,1)*A(5,5)
*           +A(1,4)*A(2,1)*A(4,2)*A(5,5)-A(1,4)*A(2,1)*A(4,5)*A(5,2)
*           -A(1,2)*A(2,4)*A(4,5)*A(5,1)+A(1,2)*A(2,4)*A(4,1)*A(5,5)
*           -A(1,2)*A(2,1)*A(4,4)*A(5,5)+A(1,2)*A(2,1)*A(4,5)*A(5,4)
*           -A(1,2)*A(2,5)*A(4,1)*A(5,4)+A(1,2)*A(2,5)*A(4,4)*A(5,1)
*           +A(1,1)*A(2,2)*A(3,3)*A(5,5)-A(1,1)*A(2,2)*A(3,5)*A(5,3)
*           +A(1,1)*A(2,5)*A(3,2)*A(5,3)-A(1,1)*A(2,5)*A(3,3)*A(5,2)
*           +A(1,1)*A(2,3)*A(3,5)*A(5,2)-A(1,1)*A(2,3)*A(3,2)*A(5,5)
*           -A(1,5)*A(2,1)*A(3,2)*A(5,3)+A(1,5)*A(2,1)*A(3,3)*A(5,2)
 F(1)=F(1)  -A(1,5)*A(2,3)*A(3,1)*A(5,2)+A(1,5)*A(2,3)*A(3,2)*A(5,1)
*           -A(1,5)*A(2,2)*A(3,3)*A(5,1)+A(1,5)*A(2,2)*A(3,1)*A(5,3)
*           +A(1,3)*A(2,5)*A(3,1)*A(5,2)-A(1,3)*A(2,5)*A(3,2)*A(5,1)
*           +A(1,3)*A(2,2)*A(3,5)*A(5,1)-A(1,3)*A(2,2)*A(3,1)*A(5,5)
*           +A(1,3)*A(2,1)*A(3,2)*A(5,5)-A(1,3)*A(2,1)*A(3,5)*A(5,2)
*           -A(1,2)*A(2,3)*A(3,5)*A(5,1)+A(1,2)*A(2,3)*A(3,1)*A(5,5)
*           -A(1,2)*A(2,1)*A(3,3)*A(5,5)+A(1,2)*A(2,1)*A(3,5)*A(5,3)
*           -A(1,2)*A(2,5)*A(3,1)*A(5,3)+A(1,2)*A(2,5)*A(3,3)*A(5,1)
*           +A(1,1)*A(2,2)*A(3,3)*A(4,4)-A(1,1)*A(2,2)*A(3,4)*A(4,3)
*           +A(1,1)*A(2,4)*A(3,2)*A(4,3)-A(1,1)*A(2,4)*A(3,3)*A(4,2)
*           +A(1,1)*A(2,3)*A(3,4)*A(4,2)-A(1,1)*A(2,3)*A(3,2)*A(4,4)
*           -A(1,4)*A(2,1)*A(3,2)*A(4,3)+A(1,4)*A(2,1)*A(3,3)*A(4,2)
*           -A(1,4)*A(2,3)*A(3,1)*A(4,2)+A(1,4)*A(2,3)*A(3,2)*A(4,1)
*           -A(1,4)*A(2,2)*A(3,3)*A(4,1)+A(1,4)*A(2,2)*A(3,1)*A(4,3)
*           +A(1,3)*A(2,4)*A(3,1)*A(4,2)-A(1,3)*A(2,4)*A(3,2)*A(4,1)
*           +A(1,3)*A(2,2)*A(3,4)*A(4,1)-A(1,3)*A(2,2)*A(3,1)*A(4,4)
*           +A(1,3)*A(2,1)*A(3,2)*A(4,4)-A(1,3)*A(2,1)*A(3,4)*A(4,2)
*           -A(1,2)*A(2,3)*A(3,4)*A(4,1)+A(1,2)*A(2,3)*A(3,1)*A(4,4)
*           -A(1,2)*A(2,1)*A(3,3)*A(4,4)+A(1,2)*A(2,1)*A(3,4)*A(4,3)
*           -A(1,2)*A(2,4)*A(3,1)*A(4,3)+A(1,2)*A(2,4)*A(3,3)*A(4,1)
```

Appendix B

Program Listing

```
C PURPOSE
C      TO PRODUCE IN LITERAL FORM (FORTRAN FORMAT) THE INVARIANT
C      COEFFICIENTS OF THE CHARACTERISTIC POLYNOMIALS FOR N-SQUARE
C      MATRICES WHERE N MAY RANGE FROM THREE TO NINE
C
C*************************************************************************
C
C
C      N  = DIMENSION OF N-SQUARE MATRIX
C
C      N1 = NUMBER OF CALLS TO COMBIN
C         = BINOMIAL COEFFICIENT (N I)
C      N2 = NUMBER OF CALLS TO PERMUT
C         = I FACTORIAL
C      N3 = NUMBER OF TERMS PER LINE (I FACOTRS PER TERM)
C         = INTEGER 8/I
C      N4 = NUMBER OF FACTORS PER LINE
C         = I*N3
C      N6 = NUMBER OF ITEMS IN COMBINATION (N THINGS TAKEN N6 AT A TIME)
C         = N-I
C      M  = MAXIMUM VALUE OF I
C         = N-1
C      M1   COUNTS CALLS TO COMBIN
C
C      M2   COUNTS CALLS TO PERMUT
C
C      M3   COUNTS NUMBER OF TERMS STORED FOR CURRENT LINE
C
C      M5   COUNTS NUMBER LINES ON CURRENT PAGE
C
C      I  = NUMBER OF FACTORS IN A TERM, NUMBER OF ITEMS BEING PERMUTED
C           COUNTS UP TO N-1=M
C      K    COUNTS UP TO I (NUMBER OF ITEMS IN PERMUTATION ARRAY)
C
C      V    COMBINATION ARRAY
C
C      LA   PERMUTATION ARRAY   LA(1,1) TO LA(I,1)  ARE THE RIGHT HAND
C           INDICES FOR THE LITERAL OUTPUT FACTORS
C      X    INITIAL PERMUTATION ITEMS   X(1) TO X(I) ARE THE LEFT HAND
C           INDICES FOR THE LITERAL OUTPUT FACTORS
C      W    THE PRINT ARRAY
C
C      ITEMS = NUMBER OF ITEMS BEING COMBINED AT A TIME
C            = N-I+1
C
C
C*************************************************************************
C
      IMPLICIT INTEGER(A-Z)
      COMMON/NP/IS(9),LA(9,9),ITEMS,IFIRST
      DIMENSION SIGN(3),V(9),W(3,8),X(9)
      DATA SIGN/'-','*','+'/
```

```
100 READ(5,1000,END=999)N
    IF( (N.LT.2).OR.N.GT.9)GO TO 999
    M=N-1
    WRITE(6,2000)N,N
    WRITE( 6,2010)
    WRITE(7,2600)N
    WRITE( 6,2100)N
    WRITE(7,2100)N
    WRITE( 6,2200)M,(( SIGN( 3),I,I ),I=1,N)
    WRITE(7,2200)M,((SIGN(3),I,I ),I=1,N)
    N1=N
    N2=1
    M5=2
    DO 200 I=2,M
    N6=N-I
    N1=(N1*(N6+1))/I
    N2=N2*I
    N3=8/I
    N4=I*N3
    ASSIGN 160 TO L
    V(1)=0
    M3=0
    M4=0
    IF( M5+(N1*N2)/N3.GT.60)M5=60
    DO 200 M1=1,N1
    CALL COMBIN(N,N6,V)
    DO 110 J=1,N
110 LA( J,1)=J
    DO 120 J=1,N6
120 LA( V(J),1)=0
    K=0
    DO 130 J=1,N
    IF(LA( J,1).EQ.0)GO TO 130
    K=K+1
    LA(K,1)=LA(J,1)
    X(K)=LA(J,1)
130 CONTINUE
    IFIRST=1
    ITEMS=I
    DO 190 M2=1,N2
    CALL PERMUT
    S=SIGN(IS(1)+2)
    DO 140 K=1,I
    W( 1,M3*I +K)=S
    W(2,M3*I +K)=X(K)
    W(3,M3*I +K)=LA(K,1)
140 S=SIGN(2)
    M3=M3+1
    IF( (M3.LT.N3).AND.(N2-M2+N1-M1.NE.0))GO TO 190
    IF( M3.LT.N3)N4=I*M3
    M3=0
    M5=M5+1
    IF(M5.LE.60) GO TO 150
    WRITE(6,2010)
    M5=1
150 GO TO L,(160,170,180)
```

```
 160 ASSIGN 180 TO L
     WRITE(6,2500)    N6,((W(J,K),J=1,3),K=1,N4)
     WRITE(7,2500)    N6,((W(J,K),J=1,3),K=1,N4)
     M7=1
     GO TO 190
 170 ASSIGN 180 TO L
     WRITE(6,2400)       N6,N6,((W(J,K),J=1,3),K=1,N4)
     WRITE(7,2400)       N6,N6,((W(J,K),J=1,3),K=1,N4)
     M7=1
     GO TO 190
 180 WRITE(6,2300)((W(J,K),J=1,3),K=1,N4)
     WRITE(7,2300)((W(J,K),J=1,3),K=1,N4)
     M7=M7+1
     IF(M7.EQ.20)ASSIGN 170 TO L
 190 CONTINUE
 200 CONTINUE
     WRITE(7,2700)
     GO TO 100
1000 FORMAT(I1)
2000 FORMAT(1H1,20(/),59X,'LITERAL FORM OF'//
    *52X,'THE INVARIANT COEFFICIENTS OF'//
    *49X,'THE CHARACTERISTIC POLYNOMIAL FOR A'///
    *64X,I1,' X ',I1///
    *64X,'MATRIX')
2010 FORMAT(1H1)
2100 FORMAT(6X,'F(',I1,')=',4X,'+1')
2200 FORMAT(6X,'F(',I1,')=',4X,8(A1,'A(',I1,',',I1,')'))
2300 FORMAT(5X,'*',9X,8(A1,'A(',I1,',',I1,')'))
2400 FORMAT(6X,'F(',I1,')=','F(',I1,')',8(A1,'A(',I1,',',I1,')'))
2500 FORMAT(6X,'F(',I1,')=',4X,8(A1,'A(',I1,',',I1,')'))
 999 STOP
2600 FORMAT('C'/'C',5X,'COEFFICIENT POLYNOMIALS FOR MATRIX OF ORDER',
    *I2/'C')
2700 FORMAT('C'/'C'/'C')
     END
```

```
      SUBROUTINE COMBIN(L,N,INDX)
C
C PURPOSE
C     TO PRODUCE THE NEXT COMBINATION (IN ASCENDING ORDER) OF L THINGS
C     TAKEN N AT A TIME.  NUMBER OF CALLS REQUIRED TO GET ALL
C     COMBINATIONS IS BINOMIAL COEFFICIENT (L N)
C
C CALLING SEQUENCE      CALL COMBIN(L,N,INDX)      WHERE
C     L      = THE NUMBER OF THINGS
C     N      = THE NUMBER OF THESE THINGS BEING COMBINED
C     INDX   = THE OUTPUT INDEX
C              INDX(1) MUST BE ZERO ON THE FIRST CALL FOR A GIVEN N, L
C              INDX(1) WILL BE SET TO ZERO AFTER THE LAST COMBINATION
C
C OUTPUT
C     THE OUTPUT IS SENT TO THE CALLING ROUTINE IN INDX.  THE FIRST
C     CALL WILL PRODUCE IN INDX(1) TO INDX(N) THE VALUES 1,2,...,N.
C     SUCCESSIVE CALLS WILL PRODUCE, IN ASCENDING ORDER, THE OTHER
C     COMBINATIONS OF THESE NUMBERS.  BY ASCENDING ORDER IS MEANT,
C     E.G., 1,2,3,4 IS LESS THAN 1,2,3,5 IS LESS THAN 1,2,4,5 ETC.
C
C SAMPLE OUTPUT FOR 11 SUCCESSIVE CALLS TO COMBIN WITH L=6, N=3,
C                AND INDX(1)=0 ON THE FIRST CALL
C     CALL NO. 1  2  3  4  5  6  7  8  9 10 11
C     INDX(1)  1  1  1  1  1  1  2  2  2  3  0
C     INDX(2)  2  2  2  3  3  4  3  3  4  4  4
C     INDX(3)  3  4  5  4  5  5  4  5  5  5  5
C
C RESTRICTIONS
C     THE VALUES OF N AND L AND THE CONTENTS OF INDX SHOULD
C     NOT BE CHANGED DURING THE (N L) CALLS TO COMBIN UNLESS A NEW
C     SEQUENCE OF COMBINATIONS IS TO BE STARTED.  IN THIS CASE, INDX(1)
C     MUST BE RESET TO 0.
C
      DIMENSION INDX(1)
      IF(INDX(1).GT.0)GO TO 110
      DO 100 I=1,N
  100 INDX(I)=I
      RETURN
  110 INDX(N)=INDX(N)+1
      IF( INDX(N).LE.L )RETURN
      L2=L
      M=N
  120 M=M-1
      IF(M.EQ.0)GO TO 140
      L2=L2-1
      INDX(M)=INDX(M)+1
      IF(INDX(M).GT.L2)GO TO 120
      J=M+1
      DO 130 I=J,N
  130 INDX(I)=INDX(I-1)+1
      RETURN
  140 INDX(1)=0
      RETURN
      END
```

```
      SUBROUTINE PERMUT
C PURPOSE
C     TO PERMUTE N  ITEMS ON N FACTORIAL CALLS TO SUBROUTINE
C
C
C METHOD
C     ON FIRST CALL TO PERMUT, IFIRST MUST EQUAL ONE, THE N ITEMS TO
C     BE PERMUTED MUST BE IN POSITIVE, ASCENDING ORDER IN
C     LA(1,1) TO LA(N,1), AND N MUST EQUAL THE NUMBER OF ITEMS BEING
C     PERMUTED.  THIS FIRST CALL IS USED TO INITIALIZE COUNTERS AND
C     THE REST OF THE LA ARRAY, AND TO SET UP THE SIGN ARRAY (IS) WITH
C     POSITIVE VALUES.
C
C
C OUTPUT
C     THE PERMUTED ITEMS WILL BE RETURNED TO THE CALLING PROGRAM IN
C     LA(1,1) TO LA(N,1) WHILE THE SIGN OF THE PERMUTATION IS
C     RETURNED IN IS(1).  THE SIGN IS INDICATED BY +1 OR -1 AS THE
C     PERMUTATION IS EVEN OR ODD.  THE FIRST CALL TO PERMUT RETURNS
C     LA(1,1) TO LA(N,1) IN THE SAME ORDER AS IT WAS RECEIVED WHILE
C     THE SIGN IS RETURNED AS +1.  HENCE THE ORIGINAL ORDER MUST BE
C     POSITIVE
C
C
C SAMPLE OUTPUT FOR N=4, LA(1,10) TO LA(4,10) = 2,4,5,7
C
C     CALL NUMBER     1  2  3  4  5  6  7  8  9 10 11 12
C     LA(1,1)         2  2  2  2  2  2  7  7  7  7  7  7
C     LA(2,1)         4  4  7  7  5  5  2  2  5  5  4  4
C     LA(3,1)         5  7  4  5  7  4  4  5  2  4  5  2
C     LA(4,1)         7  5  5  4  4  7  5  4  4  2  2  5
C     IS(1)          +1 -1 +1 -1 +1 -1 -1 +1 -1 +1 -1 +1
C
C     CALL NUMBER    13 14 15 16 17 18 19 20 21 22 23 24
C     LA(1,1)         5  5  5  5  5  5  4  4  4  4  4  4
C     LA(2,1)         7  7  4  4  2  2  5  5  2  2  7  7
C     LA(3,1)         2  4  7  2  4  7  7  2  5  7  2  5
C     LA(4,1)         4  2  2  7  7  4  2  7  7  5  5  2
C     IS(1)          +1 -1 +1 -1 +1 -1 -1 +1 -1 +1 -1 +1
C
C
C RESTRICTIONS
C     THE LA ARRAY, THE IS ARRAY, AND N MUST NOT BE CHANGED DURING THE
C     N FACTORIAL SUCCESIVE CALLS TO PERMUT UNLESS A NEW PERMUTATION
C     IS TO BE STARTED BY SETTING IFIRST BACK TO 1.  IFIRST MUST NOT BE
C     CHANGED UNLESS A NEW PERMUTATION SET IS TO BE STARTED.
C     THE ITEMS TO BE PERMUTED MUST ALL BE DIFFERENT AND, INITIALLY,
C     IN POSITIVE, ASCENDING ORDER.  OTHERWISE, THE SIGN WILL BE WRONG.
C
C
      COMMON/NP/IS(9),LA(9,9),N,IFIRST
      DIMENSION L(9),LL(9)
      GO TO (100,200),IFIRST
  100 IFIRST=2
      M=N-1
      DO 110 J=1,M
      L(J)=0
      LL(J)=J
```

```
110 IS(J)=1
    IF(N.EQ.2)RETURN
    DO 120 J=2,M
    DO 120 I=1,N
120 LA(I,J)=LA(I,1)
    RETURN
200 L(1)=L(1)+1
    IF(L(1).GT.1)GO TO 300
    K=LA(N,1)
    LA(N,1)=LA(N-1,1)
    LA(N-1,1)=K
    IS(1)=-IS(1)
    RETURN
300 M=1
410 L(M)=0
    M=M+1
    L(M)=L(M)+1
    IF(L(M).GT.LL(M))GO TO 410
    K=LA(N,M)
    DO 420 I=1,M
420 LA(N-I+1,M)=LA(N-I,M)
    M2=N-M
    LA(M2,M)=K
    M3=M-1
    IF(MOD(M,2).EQ.1)IS(M)=-IS(M)
    DO 430 J=1,M3
    IS(J)=IS(M)
    DO 430 I=M2,N
430 LA(I,J)=LA(I,M)
    RETURN
    END
```